
Private Function CheckRetType() As Boolean

'**************************************************************************

'Purpose:   Validate choice of retirement type

'           based on retirement date

'Inputs:    N/A

'Returns:   N/A

'Assumes:   Calculated values available for retirement date

'           birth date, benefit service date, adj. service date

'Effects:   Returns False and puts up message box

'           if it appears employee is not eligible

'**************************************************************************

'Revision History:

'frmRetType

'CheckRetType

'08/04/97 RG Per telcon 08/04/97, warn user if

'            "vested term" employee is also eligible

'            for early retirement

'08/03/97 RG Use exit age, not benefit commencement age

'05/29/97 RG Correct bad logic for Normal Retirement age

'**************************************************************************

Dim liRet As Integer

Const uNO_CODE As Long = 1 + vbObjectError + 512

'Validate various dates first

If Not DupRetValid Then GoTo FuncExit

'Don't want error trapping inside the VB IDE

If gbInDesign Then On Error GoTo 0

If Not gbInDesign Then On Error GoTo CheckRetError:

'Initial assumption is they don't satisfy requirements

CheckRetType = False

'Calculate age and service to test eligibility requirements

ffAgeComm = fdBenCommDate - fdBirthDate

ffAge = fdRetDate - fdBirthDate

ffElgSvc = fdRetPlsOne - fdASDate

ffBenSvc = fdRetPlsOne - fdBenSvcDate


If gbInDesign Then

    Debug.Print

    Debug.Print "Name      "; gsName

    Debug.Print "Comm Age  "; ffAgeComm

    Debug.Print "Exit Age  "; ffAge

    Debug.Print "Elig Svc  "; ffElgSvc

    Debug.Print "Benef Svc "; ffBenSvc

End If

Select Case cboRetType.Text

    

    Case gsNORM_LATE

[deleted code]

    Case gsEARLY

[deleted code]

    Case gsOPTIONAL

    'Requires age 50 with 15 years service, or

    'Age 49 + 17, age 48 + 19, etc. down to age 45 + 25

    'Based on 2 mos more service for each month earlier age

    If 2# * ffAge + ffElgSvc > 115! Then

        CheckRetType = True

    Else

        MsgBox "Age " + CStr(Round(ffAge, 2)) + _

        " and service of " + CStr(Round(ffElgSvc, 2)) + " years" + _

        vbCrLf + "Does not satisfy eligibility for " + _

        cboRetType.Text

        If gbInDesign Then

            Debug.Print "2# * ffAge + ffElgSvc= "; _

                         2# * ffAge + ffElgSvc

        End If

    End If

    Case gsINCAP

[deleted code]

    Case gsDEFER_VEST, gsDEATH_ACT

[deleted code]


    Case Else

    if gbInDesign then Stop

    Err.Raise uNO_CODE

End Select

'Maybe participant does not have 5 years service

'give user the option to quit

If ffElgSvc < 4.99999! And _

    ffAge < 64.99999! Then

    Call UserQuit(Me, _

        "This participant is not eligible for" + vbCrLf + _

        "any benefits at this retirement date!" _

        + vbCrLf + vbCrLf + _

        "Do you want to discard all calculations for this", _

        "participant, and start over with a new participant?", _

        "Participant can't receive benefits")

End If

If CheckRetType = False Then

    If txtRetDate.Visible And txtRetDate.Enabled Then 

        txtRetDate.SetFocus

    End If

End If

GoTo FuncExit


CheckRetError:

    If Err.Number = uNO_CODE Then

        gsErrMsg = "No validation code for this case"

    Else

        gsErrMsg = Err.Description

    End If

    MsgBox gsErrMsg, vbExclamation, "CheckretType Error!"

FuncExit:

End Function

Public Sub InDesign()

'**************************************************************************

'Purpose:   Set up debugging flag

'Inputs:    N/A

'Returns:   N/A

'Assumes:   N/A

'Effects:   If running program inside VB IDE, 

'           Set global variable gbInDesign to True

'           By default, value is false when program runs as EXE

'**************************************************************************

On Error GoTo eh_InDesign

'Simple test to determine when inside VB IDE

Debug.Print 1 / 0

Exit Sub

eh_InDesign:

gbInDesign = True

End Sub

NOTE 1: Must document these items for future reference (CRS)





NOTE 2: Ensure my own error numbers will not conflict with future versions of Visual Basic 








NOTE 3: Turn OFF error traps when running inside VB IDE








NOTE 4: This is “leftover” feedback from prior debugging sessions – only inside VB IDE








NOTE 5: Originally had strings for tests – much safer to declare global string constants (no typos)








NOTE 6: Every Select Case MUST have a Case Else – avoid future logic errors








NOTE 7: Plan ahead for weird situations – this code allows  user to escape from a “dead end”








NOTE 8: Be careful when relocating cursor – this can bomb if control is not enabled








NOTE 9: This is a trivial example, but EVERY routine MUST have an error handler

